
	Reason for change:
	1.	During calculation of v_TimeOfSysinfoModification in function f_NR_ModifySysinfo(), v_ProcessingDelay is added twice 

2.	In most cases, calling of f_NR_ModifySysinfo () is followed by f_NR_WaitModificationPeriods() to wait 2.1 modifciation periods to allow the UE to read the updated system information. However the current implementation will start the timer immediately on return from f_NR_ModifySysinfo(), which can be more than 1 modification period before the short message is sent from the SS, which may not leave the UE enough time to read the updated SI.

3. In function f_NR_ModifySysinfo_WaitActivation (), delay time is calculated incorrectly because the paramters in call to f_SubFrameTiming_Duration() are in the wrong order (T1 should be v_Tnow, and T2 should be v_SysInfoActivation)




	
	

	Summary of change:
	1.	Remove addition v_ProcessingDelay in calculation of  v_StartNextModPeriod_SFN (already incorporated into v_Timing_SFN)

2.	Add a delay to ensure short message is sent at approximately the time the function returns (i.e. the time v_TimeOfShortMessage is reached)

3. Reverse the arguments T1 and T2 order when calling f_SubFrameTiming_Duration()

1. [bookmark: _Toc122434485][bookmark: _Toc68621184]Overview
This document lists all the changes needed to correct issues in the ATS iWD-TTCN3-B2020-09_D21wk12 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc68621185]Corrections required

1. [bookmark: _Toc30685521][bookmark: _Toc68621186]Change 1
	Function name
	f_NR_ModifySysinfo()

	Reason for change
	1. During calculation of v_TimeOfSysinfoModification, v_ProcessingDelay is added twice 

2. In most cases, calling of f_NR_ModifySysinfo () is followed by f_NR_WaitModificationPeriods() to wait 2.1 modifciation periods to allow the UE to read the updated system information. However the current implementation will start the timer immediately on return from f_NR_ModifySysinfo(), which can be more than 1 modification period before the short message is sent from the SS, which may not leave the UE enough time to read the updated SI.

	Summary of change
	1. Remove addition v_ProcessingDelay in calculation of  v_StartNextModPeriod_SFN (already incorporated into v_Timing_SFN)

2. Add a delay to ensure short message is sent at approximately the time the function returns (i.e. the time v_TimeOfShortMessage is reached)

	TTCN module
	NR5GC_SysinfoPaging

	MCC160 Comment
	



Before Change
	function f_NR_ModifySysinfo(NR_CellId_Type p_NR_CellId,
                              boolean p_ShortMessageFlag := true,
                              NR_RRC_STATE_Type p_UE_State := RRC_IDLE,
                              boolean p_BroadcastAllSI := false
                              )
    runs on NR5GC_PTC return SubFrameTiming_Type
  {
    //Note: in INACTIVE state UE may have a different DRX cycle - that is not needed so far
    var SubFrameTiming_Type     v_Timing;
    var SubFrameTiming_Type     v_TimeOfShortMessage;
    var SubFrameTiming_Type     v_TimeOfSysinfoModification;
    var integer                 v_Timing_SFN;
    var SubFrameTiming_Type     v_StartNextModPeriod;
    var integer v_ModificationPeriodFrames := f_NR_CalculateModificationPeriod (p_NR_CellId);
    var integer v_ProcessingDelay := tsc_NR_SfnDelay; //For IDLE and INACTIVE state
    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
    var template(omit) NR_BcchToPdschConfig_Type v_NR_BcchToPdschConfig := f_NR_InitialiseBcchToPdschConfig(p_NR_CellId,p_BroadcastAllSI); //@sic R5s200703 sic@
    var integer v_StartNextModPeriod_SFN;
    
    //When UE is in connected mode, short message will be sent during a complete modification period
    //Add a longer processing delay to current SFN
    if (p_UE_State == RRC_CONNECTED and not(pc_ETWS_NR or pc_CMAS_NR)) {v_ProcessingDelay := tsc_NR_SfnDelayConnected}

    //Get the current SFN and subframe
    v_Timing := f_NR_GetCurrentTiming(p_NR_CellId);
        
    //Add processing delay to current SFN
    v_Timing_SFN := v_Timing.SFN.Number + v_ProcessingDelay;
    // Calculate the SFN of the next Modification Period
    //@sic R5-211417 sic@
    v_StartNextModPeriod_SFN := v_Timing_SFN + v_ProcessingDelay + (v_ModificationPeriodFrames -(v_Timing_SFN mod v_ModificationPeriodFrames));
    v_StartNextModPeriod.HSFN.Number := (v_Timing.HSFN.Number + (v_StartNextModPeriod_SFN/1024) ) mod 1024;
    v_StartNextModPeriod.SFN.Number :=  v_StartNextModPeriod_SFN mod 1024;
    //Set Subframe and slot to 0, i.e. start of the modification period
    v_StartNextModPeriod.Subframe.Number := 0;
    v_StartNextModPeriod := f_NR_SubFrameTiming_SetSlot(v_StartNextModPeriod, 0, v_SCS);
    
    if (not p_ShortMessageFlag) { //Short Msg Indication not to be sent
      v_TimeOfSysinfoModification := v_StartNextModPeriod;
    } else {
      //@sic R5-211417 sic@
      v_TimeOfShortMessage := v_StartNextModPeriod;
      v_StartNextModPeriod.HSFN.Number := (v_StartNextModPeriod.HSFN.Number + (v_StartNextModPeriod.SFN.Number + v_ModificationPeriodFrames)/1024) mod 1024;
      v_StartNextModPeriod.SFN.Number := (v_StartNextModPeriod.SFN.Number + v_ModificationPeriodFrames) mod 1024;
      v_TimeOfSysinfoModification := v_StartNextModPeriod;
      f_NR_ShortMessageIndication_SysinfoMod(p_NR_CellId, v_TimeOfShortMessage, p_UE_State);
    }
    f_NR_SS_ConfigureSysinfo (p_NR_CellId, cs_TimingInfo_SFNOnly(v_TimeOfSysinfoModification.SFN.Number), v_NR_BcchToPdschConfig);
    return v_TimeOfSysinfoModification;
  }



After Change
	function f_NR_ModifySysinfo(NR_CellId_Type p_NR_CellId,
                              boolean p_ShortMessageFlag := true,
                              NR_RRC_STATE_Type p_UE_State := RRC_IDLE,
                              boolean p_BroadcastAllSI := false
                              )
    runs on NR5GC_PTC return SubFrameTiming_Type
  {
    //Note: in INACTIVE state UE may have a different DRX cycle - that is not needed so far
    var SubFrameTiming_Type     v_Timing;
    var SubFrameTiming_Type     v_TimeOfShortMessage;
    var SubFrameTiming_Type     v_TimeOfSysinfoModification;
    var integer                 v_Timing_SFN;
    var SubFrameTiming_Type     v_StartNextModPeriod;
    var integer v_ModificationPeriodFrames := f_NR_CalculateModificationPeriod (p_NR_CellId);
    var integer v_ProcessingDelay := tsc_NR_SfnDelay; //For IDLE and INACTIVE state
    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
    var template(omit) NR_BcchToPdschConfig_Type v_NR_BcchToPdschConfig := f_NR_InitialiseBcchToPdschConfig(p_NR_CellId,p_BroadcastAllSI); //@sic R5s200703 sic@
    var integer v_StartNextModPeriod_SFN;
    
    //When UE is in connected mode, short message will be sent during a complete modification period
    //Add a longer processing delay to current SFN
    if (p_UE_State == RRC_CONNECTED and not(pc_ETWS_NR or pc_CMAS_NR)) {v_ProcessingDelay := tsc_NR_SfnDelayConnected}

    //Get the current SFN and subframe
    v_Timing := f_NR_GetCurrentTiming(p_NR_CellId);
        
    //Add processing delay to current SFN
    v_Timing_SFN := v_Timing.SFN.Number + v_ProcessingDelay;
    // Calculate the SFN of the next Modification Period
    //@sic R5-211417 sic@
    v_StartNextModPeriod_SFN := v_Timing_SFN + (v_ModificationPeriodFrames -(v_Timing_SFN mod v_ModificationPeriodFrames));
    v_StartNextModPeriod.HSFN.Number := (v_Timing.HSFN.Number + (v_StartNextModPeriod_SFN/1024) ) mod 1024;
    v_StartNextModPeriod.SFN.Number :=  v_StartNextModPeriod_SFN mod 1024;
    //Set Subframe and slot to 0, i.e. start of the modification period
    v_StartNextModPeriod.Subframe.Number := 0;
    v_StartNextModPeriod := f_NR_SubFrameTiming_SetSlot(v_StartNextModPeriod, 0, v_SCS);
    
    if (not p_ShortMessageFlag) { //Short Msg Indication not to be sent
      v_TimeOfSysinfoModification := v_StartNextModPeriod;
    } else {
      //@sic R5-211417 sic@
      v_TimeOfShortMessage := v_StartNextModPeriod;
      v_StartNextModPeriod.HSFN.Number := (v_StartNextModPeriod.HSFN.Number + (v_StartNextModPeriod.SFN.Number + v_ModificationPeriodFrames)/1024) mod 1024;
      v_StartNextModPeriod.SFN.Number := (v_StartNextModPeriod.SFN.Number + v_ModificationPeriodFrames) mod 1024;
      v_TimeOfSysinfoModification := v_StartNextModPeriod;
      f_NR_ShortMessageIndication_SysinfoMod(p_NR_CellId, v_TimeOfShortMessage, p_UE_State);
    }
    f_NR_SS_ConfigureSysinfo (p_NR_CellId, cs_TimingInfo_SFNOnly(v_TimeOfSysinfoModification.SFN.Number), v_NR_BcchToPdschConfig);
    if (p_ShortMessageFlag){
      // wait until the short message is to be sent 
      f_Delay(int2float(f_SubFrameTiming_Duration(v_Timing, v_TimeOfShortMessage) - 100 )/ 1000.0);
    }

    return v_TimeOfSysinfoModification;
  }




1. [bookmark: _Toc68621187]Change 2
	Function name
	f_NR_ModifySysinfo_WaitActivation ()

	Reason for change
	Delay time is calculated incorrectly because the paramters in call to f_SubFrameTiming_Duration() are in the wrong order (T1 should be v_Tnow, and T2 should be v_SysInfoActivation)

	Summary of change
	Reverse the arguments T1 and T2 order when calling f_SubFrameTiming_Duration()


	TTCN module
	NR5GC_SysinfoPaging

	MCC160 Comment
	



Before Change
	    function f_NR_ModifySysinfo_WaitActivation( NR_CellId_Type p_NR_CellId,
                                              boolean p_ShortMessageFlag := true,
                                              NR_RRC_STATE_Type p_UE_State := RRC_IDLE
                                              ) runs on NR5GC_PTC
  {
    var SubFrameTiming_Type v_SysinfoActivation;
    var SubFrameTiming_Type v_Tnow;
    
    // Modify Sysinfo and wait until it is activated
    v_SysinfoActivation := f_NR_ModifySysinfo(p_NR_CellId, p_ShortMessageFlag, p_UE_State);
    v_Tnow := f_NR_GetCurrentTiming(p_NR_CellId);

    //Wait for 1 second after the start of the modification period in which the modified system information is broadcasted to ensure UE has received Sysinfo
    f_Delay(int2float(f_SubFrameTiming_Duration(v_SysinfoActivation, v_Tnow) + 1000) / 1000.0); // KS wk12 CRxxxx //@sic R5s200678 channge 2 sic@

   }



After Change
	  function f_NR_ModifySysinfo_WaitActivation( NR_CellId_Type p_NR_CellId,
                                              boolean p_ShortMessageFlag := true,
                                              NR_RRC_STATE_Type p_UE_State := RRC_IDLE
                                              ) runs on NR5GC_PTC
  {
    var SubFrameTiming_Type v_SysinfoActivation;
    var SubFrameTiming_Type v_Tnow;
    
    // Modify Sysinfo and wait until it is activated
    v_SysinfoActivation := f_NR_ModifySysinfo(p_NR_CellId, p_ShortMessageFlag, p_UE_State);
    v_Tnow := f_NR_GetCurrentTiming(p_NR_CellId);

    //Wait for 1 second after the start of the modification period in which the modified system information is broadcasted to ensure UE has received Sysinfo
    f_Delay(int2float(f_SubFrameTiming_Duration(v_Tnow, v_SysinfoActivation) + 1000) / 1000.0); // KS wk12 CRxxxx //@sic R5s200678 channge 2 sic@

   }




